# Stabilized optimization via an NCL algorithm

Ding Ma, Kenneth L. Judd, Dominique Orban, and Michael A. Saunders

## **1** Introduction

We consider constrained optimization problems of the form

NCO	$\underset{x \in \mathbb{R}^{n}}{\text{minimize}}$	$\phi(x)$			
	subject to	$c(x) \ge 0,$	$Ax \ge b$ ,	$\ell \leq x \leq u,$	

where  $\phi(x)$  is a smooth nonlinear function,  $c(x) \in \mathbb{R}^m$  is a vector of smooth nonlinear functions, and  $Ax \ge b$  is a placeholder for a set of linear inequality or equality constraints, with *x* lying between lower and upper bounds  $\ell$  and *u*.

In some applications where  $m \gg n$ , there may be more than *n* constraints that are essentially active at a solution. The constraints do not satisfy the linear independence constraint qualification (LICQ), and general-purpose solvers are likely to have difficulty converging. Some form of regularization is required. We achieve this by adapting the augmented Lagrangian algorithm of the general-purpose optimization solver LANCELOT [4, 5, 13] to derive a sequence of regularized subproblems denoted in the next section by NC<sub>k</sub>.

Ding Ma

Management Science and Engineering, Stanford University, Stanford, CA, USA e-mail: dingma@stanford.edu

Kenneth Judd

Hoover Institution, Stanford University, Stanford, CA, USA e-mail: judd@hoover.stanford.edu

Dominique Orban

GERAD and Dept of Mathematics and Industrial Engineering, École Polytechnique, Montréal, QC, Canada

e-mail:dominique.orban@gerad.ca

#### Michael Saunders

Dept of Management Science and Engineering, Stanford University, Stanford, CA, USA e-mail: saunders@stanford.edu

### 2 BCL, LCL, and NCL methods

The theory for the large-scale solver LANCELOT is best described in terms of the general optimization problem

NECB	$\underset{x \in \mathbb{R}^{n}}{\text{minimize}}$	$\phi(x)$	
	subject to	c(x) = 0,	$\ell \leq x \leq u$

with *nonlinear equality constraints* and bounds. We let  $x^*$  denote a local solution of NECB and  $(y^*, z^*)$  denote associated multipliers. LANCELOT treats NECB by solving a sequence of *bound-constrained subproblems* of the form

BC<sub>k</sub> minimize 
$$L(x, y_k, \rho_k) = \phi(x) - y_k^T c(x) + \frac{1}{2} \rho_k ||c(x)||^2$$
  
subject to  $\ell \le x \le u$ ,

where  $y_k$  is an estimate of the Lagrange multipliers  $y^*$  for the equality constraints. This was called a bound-constrained Lagrangian (BCL) method by Friedlander and Saunders [8], in contrast to the LCL (linearly constrained Lagrangian) methods of Robinson [16] and MINOS [14], whose subproblems LC<sub>k</sub> contain bounds as in BC<sub>k</sub> and also linearizations of the equality constraints at the current point  $x_k$  (including linear constraints).

In order to treat NCO with a sequence of  $BC_k$  subproblems, we convert the nonlinear inequality constraints to equalities to obtain

NCO' minimize  $\phi(x)$ subject to c(x) - s = 0,  $Ax \ge b$ ,  $\ell \le x \le u$ ,  $s \ge 0$ 

with corresponding subproblems (including linear constraints)

BC<sub>k</sub>' minimize 
$$L(x, y_k, \rho_k) = \phi(x) - y_k^T(c(x) - s) + \frac{1}{2}\rho_k ||c(x) - s||^2$$
  
subject to  $Ax \ge b$ ,  $\ell \le x \le u$ ,  $s \ge 0$ .

We now introduce variables r = -(c(x) - s) into  $BC_k'$  to obtain the *nonlinearly constrained Lagrangian* (NCL) subproblem

NC<sub>k</sub> minimize 
$$\phi(x) + y_k^T r + \frac{1}{2}\rho_k ||r||^2$$
  
subject to  $c(x) + r \ge 0$ ,  $Ax \ge b$ ,  $\ell \le x \le u$ ,

in which *r* serves to make the nonlinear constraints independent. Assuming existence of finite multipliers and feasibility, for  $\rho_k > 0$  and larger than a certain finite value, the NCL subproblems should cause  $y_k$  to approach  $y^*$  and most of the solution  $(x_k^*, r_k^*, y_k^*, z_k^*)$  of NC<sub>k</sub> to approach  $(x^*, y^*, z^*)$ , with  $r_k^*$  approaching zero.

Problem NC<sub>k</sub> is analogous to Friedlander and Orban's formulation for convex quadratic programs [7, Eq. (3.2)]. See also Arreckx and Orban [2], where the motivation is the same as here, achieving reliability when the nonlinear constraints don't satisfy LICQ.

Note that for general problems NECB, the BCL and LCL subproblems contain linear constraints (bounds only, or linearized constraints and bounds). Our NCL formulation retains nonlinear constraints in the NC<sub>k</sub> subproblems, but simplifies them by ensuring that they satisfy LICQ. On large problems, the additional variables  $r \in \mathbb{R}^m$  in NC<sub>k</sub> may be detrimental to active-set solvers like MINOS or SNOPT [9] because they increase the number of degrees of freedom (superbasic variables). Fortunately they are easily accommodated by interior methods, as our numerical results show for IPOPT [17, 10]. We trust that the same will be true for KNITRO [3, 12].

#### 2.1 The BCL algorithm

The LANCELOT BCL method is summarized in Algorithm BCL. Each subproblem BC<sub>k</sub> is solved with a specified optimality tolerance  $\omega_k$ , generating an iterate  $x_k^*$  and the associated Lagrangian gradient  $z_k^* \equiv \nabla L(x_k^*, y_k, \rho_k)$ . If  $||c(x_k^*)||$  is sufficiently small, the iteration is regarded as "successful" and an update to  $y_k$  is computed from  $x_k^*$ . Otherwise,  $y_k$  is not altered but  $\rho_k$  is increased.

Key properties are that the subproblems are solved inexactly, the penalty parameter is increased only finitely often, and the multiplier estimates  $y_k$  need not be assumed bounded. Under certain conditions, all iterations are eventually successful, the  $\rho_k$ 's remain constant, the iterates converge superlinearly, and the algorithm terminates in a finite number of iterations [4].

Note that at step 8 of Algorithm BCL, the inexact minimization would be typically carried out from the initial guess  $(x_k^*, z_k^*)$ . However, other initial points are possible. At step 12, we say that  $(x_k, y_k, z_k)$  solves NECB to within  $\omega_*$  if the largest dual infeasibility is smaller than  $\omega_*$ .

### 2.2 The NCL algorithm

To derive a stabilized algorithm for problem NCO, we modify Algorithm BCL by introducing *r* and replacing the subproblems BC<sub>k</sub> by NC<sub>k</sub>. The resulting method is summarized in Algorithm NCL. The update to  $y_k$  becomes  $y_k^* \leftarrow y_k - \rho_k(c(x_k^*) - s_k^*) = y_k + \rho_k r_k^*$ , the value satisfied by an optimal  $y_k^*$  for subproblem NC<sub>k</sub>. Step 8 of Algorithm NCL would typically use  $(x_k^*, r_k^*, y_k^*, z_k^*)$  as initial guess, and that is what we use in our implementation below.

Algorithm 1 BCL (Bound-Constrained Lagrangian Method for NECB)

1: **procedure** BCL( $x_0$ ,  $y_0$ ,  $z_0$ ) 2: Set penalty parameter  $\rho_1 > 0$ , scale factor  $\tau > 1$ , and constants  $\alpha, \beta > 0$  with  $\alpha < 1$ . 3: Set positive convergence tolerances  $\eta_*, \omega_* \ll 1$  and infeasibility tolerance  $\eta_1 > \eta_*$ . 4:  $k \leftarrow 0$ , converged  $\leftarrow$  false 5: repeat  $k \gets k+1$ 6: 7: Choose optimality tolerance  $\omega_k > 0$  such that  $\lim_{k\to\infty} \omega_k \le \omega_*$ . Find  $(x_k^*, z_k^*)$  that solves BC<sub>k</sub> to within  $\omega_k$ . 8: 9: if  $||c(x_k^*)|| \le \max(\eta_*, \eta_k)$  then 10:  $y_k^* \leftarrow y_k - \rho_k c(x_k^*)$  $x_k \leftarrow x_k^*, y_k \leftarrow y_k^*, z_k \leftarrow z_k^*$ 11: if  $(x_k, y_k, z_k)$  solves NECB to within  $\omega_*$ , converged  $\leftarrow$  true 12:  $\rho_{k+1} \leftarrow \rho_k$ 13: 14:  $\eta_{k+1} \leftarrow \eta_k/(1+\rho_{k+1}^\beta)$ 15: else  $\rho_{k+1} \leftarrow \tau \rho_k$ 16: 17:  $\eta_{k+1} \leftarrow \eta_0/(1+\rho_{k+1}^{\alpha})$ end if 18: 19: until converged  $x^* \leftarrow x_k, \ y^* \leftarrow y_k, \ z^* \leftarrow z_k$ 20: 21: end procedure

#### Algorithm 2 NCL (Nonlinearly Constrained Lagrangian Method for NCO)

1: **procedure** NCL( $x_0, r_0, y_0, z_0$ ) 2: Set penalty parameter  $\rho_1 > 0$ , scale factor  $\tau > 1$ , and constants  $\alpha, \beta > 0$  with  $\alpha < 1$ . 3: Set positive convergence tolerances  $\eta_*, \omega_* \ll 1$  and infeasibility tolerance  $\eta_1 > \eta_*$ . 4:  $k \leftarrow 0$ , converged  $\leftarrow$  false 5: repeat 6:  $k \gets k+1$ 7: Choose optimality tolerance  $\omega_k > 0$  such that  $\lim_{k\to\infty} \omega_k \le \omega_*$ . 8: Find  $(x_k^*, r_k^*, y_k^*, z_k^*)$  that solves NC<sub>k</sub> to within  $\omega_k$ . 9: if  $||r_k^*|| \leq \max(\eta_*, \eta_k)$  then 10:  $y_k^* \leftarrow y_k + \rho_k r_k^*$  $x_k \leftarrow x_k^*, \ r_k \leftarrow r_k^*, \ y_k \leftarrow y_k^*, \ z_k \leftarrow z_k^*$ 11: if  $(x_k, y_k, z_k)$  solves NCO to within  $\hat{\omega}_*$ , converged  $\leftarrow$  true 12: 13:  $\rho_{k+1} \leftarrow \rho_k$  $\eta_{k+1} \leftarrow \eta_k/(1+\rho_{k+1}^\beta)$ 14: else 15:  $\rho_{k+1} \leftarrow \tau \rho_k$ 16: 17:  $\eta_{k+1} \leftarrow \eta_0 / (1 + \rho_{k+1}^{\alpha})$ end if 18: 19: until converged  $x^* \leftarrow x_k, \ r^* \leftarrow r_k, \ y^* \leftarrow y_k, \ z^* \leftarrow z_k$ 20: 21: end procedure

#### **3** An application: optimal tax policy

Some challenging test cases arise from the tax policy models described in [11]. With x = (c, y), they take the form

TAX maximize 
$$\sum_{i,y} \sum_{i} \lambda_i U^i(c_i, y_i)$$
  
subject to  $U^i(c_i, y_i) - U^i(c_j, y_j) \ge 0$  for all  $i, j$   
 $\lambda^T(y-c) \ge 0$   
 $c, y \ge 0$ ,

where  $c_i$  and  $y_i$  are the consumption and income of taxpayer *i*, and  $\lambda$  is a vector of positive weights. The utility functions  $U^i(c_i, y_i)$  are each of the form

$$U(c,y) = \frac{(c-\alpha)^{1-1/\gamma}}{1-1/\gamma} - \psi \frac{(y/w)^{1/\eta+1}}{1/\eta+1}$$

where w is the wage rate and  $\alpha$ ,  $\gamma$ ,  $\psi$  and  $\eta$  are taxpayer heterogeneities. More precisely, the utility functions are of the form

$$U^{i,j,k,g,h}(c_{p,q,r,s,t}, y_{p,q,r,s,t}) = \frac{(c_{p,q,r,s,t} - \alpha_k)^{1-1/\gamma_h}}{1 - 1/\gamma_h} - \psi_g \frac{(y_{p,q,r,s,t}/w_i)^{1/\eta_j+1}}{1/\eta_j + 1}$$

where (i, j, k, g, h) and (p, q, r, s, t) run over *na* wage types, *nb* elasticities of labor supply, *nc* basic need types, *nd* levels of distaste for work, and *ne* elasticities of demand for consumption, with *na*, *nb*, *nc*, *nd*, *ne* determining the size of the problem, namely m = T(T-1) nonlinear constraints, n = 2T variables, with  $T := na \times nb \times$  $nc \times nd \times ne$ .

Table 1 summarizes results for a 4D example (ne = 1 and  $\gamma_1 = 1$ ). The first term of U(c, y) becomes  $\log(c - \alpha)$ , the limit as  $\gamma \rightarrow 1$ . Problem NCO and Algorithm NCL were formulated in the AMPL modeling language [6]. The solvers SNOPT [9] and IPOPT [17] were unable to solve NCO itself, but Algorithm NCL was successful with IPOPT solving the subproblems NC<sub>k</sub>. We use a default configuration of IPOPT with MUMPS [1] as symmetric indefinite solver to compute search directions. We set the optimality tolerance for IPOPT to  $\omega_k = 10^{-6}$  throughout, and specified warm starts for  $k \ge 2$  using options warm\_start\_init\_point=yes and mu\_init=1e-4. These options greatly improved the performance of IPOPT on each subproblem compared to cold starts, for which mu\_init=0.1. It is helpful that only the objective function of NC<sub>k</sub> changes with k.

For this example, problem NCO has m = 39006 nonlinear inequality constraints and one linear constraint in n = 395 variables x = (c, y), and nonnegativity bounds. Subproblem NC<sub>k</sub> has 39007 constraints and 39402 variables when r is included. Fortunately r does not affect the complexity of each IPOPT iteration, but greatly improves stability. In contrast, active-set methods like MINOS and SNOPT are very inefficient on the NC<sub>k</sub> subproblems because the large number of inequality

Stabilized optimization via an NCL algorithm

k	$\rho_k$	$\eta_k$	$\ r_k^*\ _{\infty}$	$\phi(x_k^*)$	Itns	Time
1	$10^{2}$	$10^{-2}$	3.1e-03	-2.1478532e+01	125	42.8
2	$10^{2}$	$10^{-3}$	1.3e-03	-2.1277587e+01	18	6.5
3	$10^{3}$	$10^{-3}$	6.6e-04	-2.1177152e+01	27	9.1
4	$10^{3}$	$10^{-4}$	5.5e-04	-2.1110210e+01	31	10.8
5	$10^{4}$	$10^{-4}$	2.9e-04	-2.1066664e+01	57	24.3
6	$10^{5}$	$10^{-4}$	6.5e-05	-2.1027152e+01	75	26.8
7	$10^{5}$	$10^{-5}$	5.2e-05	-2.1018896e+01	130	60.9
				-2.1015295e+01		
9	$10^{6}$	$10^{-6}$	2.0e-06	-2.1014808e+01	139	70.0
10	$10^{7}$	$10^{-6}$	2.1e-07	-2.1014800e+01	177	97.6

**Table 1** NCL results on a 4D example with na, nb, nc, nd = 11, 3, 3, 2, giving m = 39006, n = 395. Itns refers to IPOPT's primal-dual interior point method, and Time is seconds on an Apple iMac with 2.93 GHz Intel Core i7.

$k \rho_k$	$\eta_k$	$\ r_k^*\ _{\infty}$	$\phi(x_k^*)$	Itns	Time
$1 \ 10^2$	$10^{-2}$	7.0e-03	-4.2038075e+02	95	41.1
$2 \ 10^2$	$10^{-3}$	4.1e-03	-4.2002898e+02	17	7.2
3 10 <sup>3</sup>	$10^{-3}$	1.3e-03	-4.1986069e+02	20	8.1
			-4.1972958e+02		
$5 \ 10^4$	$10^{-4}$	2.2e-04	-4.1968646e+02	43	20.5
$6 \ 10^5$	$10^{-4}$	9.8e-05	-4.1967560e+02	64	32.9
			-4.1967177e+02		
			-4.1967150e+02		
9 10 <sup>6</sup>	$10^{-6}$	9.4e-07	-4.1967138e+02	96	53.6

Table 2 NCL results on a 5D example with na, nb, nc, nd, ne = 5, 3, 3, 2, 2, giving m = 32220, n = 360.

constraints leads to thousands of minor iterations, and the presence of r (with no bounds) leads to thousands of superbasic variables. About 3.2n constraints were within  $10^{-6}$  of being active.

Table 2 summarizes results for a 5D example. The NC<sub>k</sub> subproblems have m = 32220 nonlinear constraints and n = 360 variables, leading to 32581 variables including *r*. Again the options warm\_start\_init\_point=yes and mu\_init=1e-4 for  $k \ge 2$  led to good performance by IPOPT on each subproblem. About 3*n* constraints were within  $10^{-6}$  of being active.

For much larger problems of this type, we found that it was helpful to reduce mu\_init more often, as illustrated in Table 3. The NC<sub>k</sub> subproblems here have m = 570780 nonlinear constraints and n = 1512 variables, leading to 572292 variables including *r*. Note that the number of NCL iterations is stable ( $k \le 10$ ), and IPOPT performs well on each subproblem with decreasing mu\_init. This time about 6.6*n* constraints were within  $10^{-6}$  of being active.

Note that the LANCELOT approach allows early subproblems to be solved less accurately. It may save time to set  $\omega_k = \eta_k$  (say) rather than  $\omega_k = \omega_*$  throughout.

Stabilized optimization via an NCL algorithm

February 10, 2018

k	$\rho_k$	$\eta_k$	$\ r_k^*\ _{\infty}$	$\phi(x_k^*)$	mu_init	Itns	Time
1	$10^{2}$	$10^{-2}$	5.1e-03	-1.7656816e+03	$10^{-1}$	825	7763.3
2	$10^{2}$	$10^{-3}$	2.4e-03	-1.7648480e+03	$10^{-4}$	66	472.8
3	$10^{3}$	$10^{-3}$	1.3e-03	-1.7644006e+03	$10^{-4}$	106	771.3
4	$10^{4}$	$10^{-3}$	3.8e-04	-1.7639491e+03	$10^{-5}$	132	1347.0
5	$10^{4}$	$10^{-4}$	3.2e-04	-1.7637742e+03	$10^{-5}$	229	2450.9
6	$10^{5}$	$10^{-4}$	8.6e-05	-1.7636804e+03	$10^{-6}$	104	1096.9
7	$10^{5}$	$10^{-5}$	4.9e-05	-1.7636469e+03	$10^{-6}$	143	1633.4
8	$10^{6}$	$10^{-5}$	1.5e-05	-1.7636252e+03	$10^{-7}$	71	786.1
9	$10^{7}$	$10^{-5}$	2.8e-06	-1.7636196e+03	$10^{-7}$	67	725.7
10	$10^{7}$	$10^{-6}$	5.1e-07	-1.7636187e+03	$10^{-8}$	18	171.0

**Table 3** NCL results on a 5D example with na, nb, nc, ne, ne = 21, 3, 3, 2, 2, giving m = 570780, n = 1512.

# 4 Conclusions

This work has been illuminating in several ways as we sought to improve our ability to solve examples of problem TAX.

- Small examples of the tax model solve efficiently with MINOS and SNOPT, but eventually fail to converge as the problem size increases.
- IPOPT also solves small examples efficiently, but eventually starts requesting additional memory for the MUMPS sparse linear solver. The solver may freeze, or the iterations may diverge.
- The NC<sub>k</sub> subproblems are not suitable for MINOS or SNOPT because of the large number of variables (*x*, *r*) and the resulting number of superbasic variables (although warm-starts are natural).
- It is often said that interior methods cannot be warm-started. Nevertheless, IPOPT has several runtime options that have proved to be extremely helpful for implementing Algorithm NCL. For the results obtained here, it has been sufficient to say that warm starts are wanted for k > 1, and that the IPOPT barrier parameter should be initialized at decreasing values for later k (where only the objective of subproblem NC<sub>k</sub> changes with k).
- The numerical examples of section 3 had 3*n*, 3*n* and 6.6*n* constraints essentially active at the solution, yet were solved successfully. They suggest that the NCL approach with an interior method as subproblem solver can overcome LICQ difficulties on problems that could not be solved directly.

### Funding

This work was supported by the National Institute of General Medical Sciences of the National Institutes of Health [award U01GM102098] (DM and MAS), XXX (KLJ), and an NSERC Discovery Grant (DO).

Acknowledgements We are extremely grateful to the developers of AMPL and IPOPT for making the development and evaluation of Algorithm NCL possible. We are especially grateful to Mehiddin Al-Baali and other organizers of the NAO-IV conference *Numerical Analysis and Optimization* at Sultan Qaboos University, Muscat, Oman, which brought the authors and AMPL developers together in January 2017.

#### References

- Patrick R. Amestoy, Iain S. Duff, Jean-Yves L'Excellent, and Jacko Koster. A fully asynchronous multifrontal solver using distributed dynamic scheduling. *SIAM Journal on Matrix Analysis and Applications*, 23(1):15–41, 2001.
- S. Arreckx and D. Orban. A regularized factorization-free method for equality-constrained optimization. Technical Report GERAD G-2016-65, GERAD, Montréal, QC, Canada, 2016.
- Richard H. Byrd, Jorge Nocedal, and Richard A. Waltz. Knitro: An integrated package for nonlinear optimization. In G. Di Pillo and M. Roma, editors, *Large-Scale Nonlinear Optimization*, pages 35–59. Springer US, Boston, MA, 2006.
- A. R. Conn, N. I. M. Gould, and Ph. L. Toint. A globally convergent augmented Lagrangian algorithm for optimization with general constraints and simple bounds. *SIAM J. Numer. Anal.*, 28:545–572, 1991.
- A. R. Conn, N. I. M. Gould, and Ph. L. Toint. LANCELOT: A Fortran Package for Large-scale Nonlinear Optimization (Release A). Lecture Notes in Computation Mathematics 17. Springer Verlag, Berlin, Heidelberg, New York, London, Paris and Tokyo, 1992.
- 6. R. Fourer, D. M. Gay, and B. W. Kernighan. *AMPL: A Modeling Language for Mathematical Programming.* Brooks/Cole, Pacific Grove, second edition, 2002.
- M. P. Friedlander and D. Orban. A primal-dual regularized interior-point method for convex quadratic programs. *Math. Prog. Comp.*, 4(1):71–107, 2012.
- M. P. Friedlander and M. A. Saunders. A globally convergent linearly constrained Lagrangian method for nonlinear optimization. *SIAM J. Optim.*, 15(3):863–897, 2005.
- P. E. Gill, W. Murray, and M. A. Saunders. SNOPT: An SQP algorithm for large-scale constrained optimization. *SIAM Review*, 47(1):99–131, 2005. SIGEST article.
- 10. IPOPT open source NLP solver. https://projects.coin-or.org/Ipopt.
- 11. K. L. Judd, D. Ma, M. A. Saunders, and C.-L. Su. Optimal income taxation with multidimensional taxpayer types. Working paper, Hoover Institution, Stanford University, 2017.
- KNITRO optimization software. https://www.artelys.com/tools/knitro\_doc/ 2\_userGuide.html.
- LANCELOT optimization software. http://www.numerical.rl.ac.uk/ lancelot/blurb.html.
- B. A. Murtagh and M. A. Saunders. A projected Lagrangian algorithm and its implementation for sparse nonlinear constraints. *Math. Program. Study*, 16:84–117, 1982.
- 15. NCL. http://stanford.edu/group/SOL/multiscale/models/NCL/.
- S. M. Robinson. A quadratically-convergent algorithm for general nonlinear programming problems. *Math. Program.*, 3:145–156, 1972.
- 17. A. Wächter and L. T. Biegler. On the implementation of a primal-dual interior point filter line search algorithm for large-scale nonlinear programming. *Math. Program.*, 106(1), 2006.

### Appendix A AMPL models, data, and scripts

Algorithm NCL has been implemented in the AMPL modeling language [6] and tested on problem TAX. The following sections list each relevant file. The files are available from [15].

### A.1 Tax model

File pTax5Dncl.mod codes subproblem NC<sub>k</sub> for problem TAX with five parameters w,  $\eta$ ,  $\alpha$ ,  $\psi$ ,  $\gamma$ , using  $\mu := 1/\eta$ . Note that for U(c, y) in the objective and constraint functions, the first term  $(c - \alpha)^{1-1/\gamma}/(1-1/\gamma)$  is replaced by a piecewise-smooth function that is defined for all values of c and  $\alpha$  (see [11]).

Primal regularization  $\frac{1}{2}\delta ||(c, y)||^2$  with  $\delta = 10^{-8}$  is added to the objective function to promote uniqueness of the minimizer. The vector *r* is called R to avoid a clash with subscript r.

```
# pTax5Dncl.mod
```

```
# Define parameters for agents (taxpayers)
param na;
                                   # number of types in wage
param nb;  # number of types in eta
param nc;  # number of types in alpha
param nd;  # number of types in psi
param ne;  # number of types in gamma
set A := 1..na;  # set of wages
set B := 1..nb;  # set of eta
set C := 1..nc;  # set of alpha
set D := 1..nd;  # set of psi
set E := 1..ne;  # set of gamma
set T = {A,B,C,D,E};  # set of agents
param nb;
                                    # number of types in eta
# Define wages for agents (taxpayers)
param wmin; # minimum wage level
param wmax;
                                    # maximum wage level
                         # maximum wage le
# i, wage vector
param w {A};
param mu{B};
param mu{B};  # j, mu = l/eta# mu vector
param mul{B};  # mul[j] = mu[j] + 1
param alpha{C};  # k, ak vector for utility
"
                                   # j, mu = 1/eta# mu vector
param psi{D}; # g
param gamma{E}; # h
param lambda{A,B,C,D,E}; # distribution density
param epsilon;
                        default 1e-8; # Small primal regularization
param primreg
var c{(i,j,k,g,h) in T} >= 0.1; # consumption for tax payer (i,j,k,g,h)
var y\{(i, j, k, g, h) \text{ in } T\} \ge 0.1; \# \text{ income}
                                                                  for tax payer (i,j,k,g,h)
var R\{(i,j,k,g,h) \text{ in } T, (p,q,r,s,t) \text{ in } T\}
         !(i=p and j=q and k=r and g=s and h=t)} >= -1e+20, <= 1e+20;
```

```
param kmax
                 default 20;
                                      # limit on NCL itns
param rhok
                 default 1e+2;
                                      # augmented Lagrangian penalty parameter
                default 10.0;
param rhofac
                                      # increase factor
                 default 1e+8;
param rhomax
                                      # biggest rhok
                                  # biggest rh
# opttol for
# reduction
# smallest e
# max r (for
# min r (for
# ||r||_inf
# guit if bi
param etak
                default 1e-2;
                                      # opttol for augmented Lagrangian loop
param etafac
                default 0.1;
                                      # reduction factor for opttol
                                     # smallest etak
                default 1e-8;
param etamin
                default 0;
                                     # max r (for printing)
param rmax
param rmin
               default
                             0;
                                     # min r (for printing)
               default 0;
param rnorm
param rtol
                default 1e-6;
                                     # quit if biggest |r_i| <= rtol</pre>
                                      # nT = na*nb*nc*nd*ne
param nT
                 default
                            1;
param m
                 default
                            1;
                                      # nT*(nT-1) = no. of nonlinear constraints
                                                   = no. of nonlinear variables
param n
                 default
                            1;
                                      # 2*nT
param ck{(i,j,k,g,h) in T} default 0;
                                               # current variable c
param yk{(i,j,k,g,h) in T} default 0;
                                              # current variable v
param rk\{(i, j, k, g, h) \text{ in } T, (p, q, r, s, t) \text{ in } T: # current variable <math>r = -(c(x) - s)
   !(i=p and j=q and k=r and g=s and h=t)} default 0;
param dk{(i,j,k,g,h) in T, (p,q,r,s,t) in T: \# current dual variables (y\_k)
   !(i=p and j=q and k=r and g=s and h=t)} default 0;
minimize f:
   sum{(i,j,k,g,h) in T}
   (
      (if c[i, j, k, g, h] - alpha[k] >= epsilon then
           - lambda[i,j,k,g,h] *
                ((c[i,j,k,g,h] - alpha[k])^(1-1/gamma[h]) / (1-1/gamma[h])
                 - psi[g]*(y[i,j,k,g,h]/w[i])^mu1[j] / mu1[j])
       else
           - lambda[i,j,k,g,h] *
             0.5/gamma[h] * epsilon^(-1/gamma[h]-1) * (c[i,j,k,g,h] - alpha[k])^2
          ( –
           + ( 1+1/gamma[h]) * epsilon^(-1/gamma[h] ) * (c[i,j,k,g,h] - alpha[k])
          + (1/(1-1/gamma[h]) - 1 - 0.5/gamma[h]) * epsilon^(1-1/gamma[h])
                 - psi[g]*(y[i,j,k,g,h]/w[i])^mu1[j] / mu1[j])
      )
   + 0.5 * primreg * (c[i,j,k,g,h]<sup>2</sup> + y[i,j,k,g,h]<sup>2</sup>)
   )
 + sum{(i,j,k,g,h) in T, (p,q,r,s,t) in T: !(i=p and j=q and k=r and g=s and h=t)}
        (dk[i,j,k,g,h,p,q,r,s,t] * R[i,j,k,g,h,p,q,r,s,t]
                    + 0.5 * rhok * R[i,j,k,g,h,p,q,r,s,t]^2);
subject to
Incentive{(i,j,k,g,h) in T, (p,q,r,s,t) in T:
          !(i=p and j=q and k=r and g=s and h=t)}:
   (if c[i, j, k, g, h] - alpha[k] >= epsilon then
      (c[i,j,k,g,h] - alpha[k])^(1-1/gamma[h]) / (1-1/gamma[h])
       - psi[g] * (y[i, j, k, g, h]/w[i]) ^mu1[j] / mu1[j]
    else
          0.5/gamma[h] *epsilon^(-1/gamma[h]-1)*(c[i,j,k,g,h] - alpha[k])^2
       + (1+1/gamma[h])*epsilon^(-1/gamma[h])*(c[i,j,k,g,h] - alpha[k])
```

```
Stabilized optimization via an NCL algorithm
                                      February 10, 2018
                                                                  11
         + (1/(1-1/gamma[h]) - 1 - 0.5/gamma[h]) *epsilon^(1-1/gamma[h])
         - psi[g]*(y[i,j,k,g,h]/w[i])^mu1[j] / mu1[j]
     )
   - (if c[p,q,r,s,t] - alpha[k] \ge epsilon then
        (c[p,q,r,s,t] - alpha[k])^{(1-1/gamma[h])} / (1-1/gamma[h])
         - psi[g]*(y[p,q,r,s,t]/w[i])^mu1[j] / mu1[j]
      else
         - 0.5/gamma[h] *epsilon^(-1/gamma[h]-1)*(c[p,q,r,s,t] - alpha[k])^2
         + (1+1/gamma[h])*epsilon^(-1/gamma[h])*(c[p,q,r,s,t] - alpha[k])
         + (1/(1-1/gamma[h]) - 1 - 0.5/gamma[h])*epsilon^(1-1/gamma[h])
         - psi[g] * (y[p,q,r,s,t]/w[i]) ^mu1[j] / mu1[j]
     )
   + R[i,j,k,g,h,p,q,r,s,t] >= 0;
  Technology:
     sum{(i,j,k,g,h) in T} lambda[i,j,k,g,h]*(y[i,j,k,g,h] - c[i,j,k,g,h]) >= 0;
```

# A.2 Tax model data

File pTax5Dncl.dat provides data for a specific problem.

```
# pTax5Dncl.dat
data;
let na := 5;
let nb := 3;
let nc := 3;
let nd := 2;
let ne := 2;
# Set up wage dimension intervals
let wmin := 2;
let wmax := 4;
let {i in A} w[i] := wmin + ((wmax-wmin)/(na-1))*(i-1);
data;
param mu :=
       0.5
   1
    2
        1
    3
        2;
# Define mul
let {j in B} mu1[j] := mu[j] + 1;
data;
param alpha :=
   1
      0
```

```
2
      1
    3
      1.5;
param psi :=
   1 1
    2
       1.5;
param gamma :=
   1 2
   2
      3;
# Set up 5 dimensional distribution
let {(i,j,k,g,h) in T} lambda[i,j,k,g,h] := 1;
# Choose a reasonable epsilon
let epsilon := 0.1;
```

### A.3 Initial values

File pTax5Dinitial.run solves a simplified model to compute starting values for Algorithm NCL. The nonlinear inequality constraints are removed, and y = c is enforced. This model solves easily with MINOS or SNOPT on all cases tried. Solution values are output to file p5Dinitial.dat.

```
# pTax5Dinitial.run
# number of types in wage
# number of types in eta
param nc := 3; # number of types in alpha
param nd := 2; # number of types in psi
param ne := 2; # number of types '-
set A := 1..na;
# Define parameters for agents (taxpayers)
set B := 1..nb;
                            # set of eta
set C := 1..nc;
                            # set of alpha
set D := 1..nd;  # set of psi
set E := 1..ne;  # set of gamma
set T = \{A, B, C, D, E\};
                              # set of agents
# Define wages for agents (taxpayers)
param wmin := 2;  # minimum wage level
                                # maximum wage level
param wmax := 4;
param w {i in A} := wmin + ((wmax-wmin)/(na-1))*(i-1); # wage vector
# Choose a reasonable epsilon
param epsilon := 0.1;
# mu vector
                     # mu = 1/eta
param mu {B};
```

```
Stabilized optimization via an NCL algorithm
                                     February 10, 2018
                                                                13
  param mu1{B};
                            # mu1[j] = mu[j] + 1
  param alpha {C};
  param gamma {E};
  param psi {D};
  var c {(i,j,k,g,h) in T} >= 0.1;
  var y {(i,j,k,g,h) in T} >= 0.1;
  maximize f: sum{(i,j,k,g,h) in T}
     if c[i, j, k, g, h] - alpha[k] >= epsilon then
       (c[i,j,k,g,h] - alpha[k])^(1-1/gamma[h]) / (1-1/gamma[h])
        - psi[g] * (y[i,j,k,g,h]/w[i])^mu1[j] / mu1[j]
     else
        - 0.5/gamma[h] *epsilon^(-1/gamma[h]-1)*(c[i,j,k,g,h] - alpha[k])^2
        + (1+1/gamma[h])*epsilon^(-1/gamma[h]) *(c[i,j,k,g,h] - alpha[k])
        + (1/(1-1/gamma[h]) -1 - 0.5/gamma[h])*epsilon^(1-1/gamma[h])
        - psi[g] * (y[i,j,k,g,h]/w[i])^mu1[j] / mu1[j];
  subject to
     Budget {(i,j,k,g,h) in T}: y[i,j,k,g,h] - c[i,j,k,g,h] = 0;
  let {(i,j,k,g,h) in T} y[i,j,k,g,h] := i+1;
  let {(i,j,k,g,h) in T} c[i,j,k,g,h] := i+1;
  data;
  param mu :=
     1
         0.5
      2
         1
      3
         2;
  # Define mul
  let {j in B} mu1[j] := mu[j] + 1;
  data;
  param alpha :=
     1 0
      2 1
      3
         1.5;
  param psi :=
     1 1
         1.5;
      2
```

param gamma := 1 2

3;

option solver snopt; option show\_stats 1;

option snopt\_options '  $\$ summary\_file=6

\

Stabilized optimization via an NCL algorithm

```
print_file=9
                           ١
   scale=no
                           \
   print_level=0
   major_iterations=2000\
   iterations=50000
                           \
   optimality_tol=1e-7
                          \backslash
  *penalty=100.0
   superbasics_limit=3000\
   solution=yes
                           \
  *verify_level=3
                           \backslash
';
display na, nb, nc, nd, ne;
solve;
display na, nb, nc, nd, ne;
display y,c >p5Dinitial.dat;
close p5Dinitial.dat;
```

### A.4 NCL implementation

File pTax5Dnclipopt.run uses files

pTax5Dinitial.run pTax5Dncl.mod pTax5Dncl.dat pTax5Dinitial.dat

to implement Algorithm NCL. Subproblems NC<sub>k</sub> are solved in a loop until  $||r_k^*||_{\infty} \le$ rtol = 1e-6, or  $\eta_k$  has been reduced to parameter etamin = 1e-8, or  $\rho_k$  has been increased to parameter rhomax = 1e+8. The loop variable k is called K to avoid a clash with subscript k in the model file.

Optimality tolerance  $\omega_k = 10^{-6}$  is used throughout to ensure that the solution of the final subproblem NC<sub>k</sub> will be close to a solution of the original problem if  $||r_k^*||_{\infty}$  is small enough for the final k ( $||r_k^*||_{\infty} \le \text{rtol} = 1\text{e}-6$ ).

IPOPT is used to solve each subproblem  $NC_k$ , with runtime options set to implement increasingly warm starts.

```
# pTax5Dnclipopt.run
reset;
model pTax5Dinitial.run;
reset;
model pTax5Dncl.mod;
data pTax5Dncl.dat;
data; var include p5Dinitial.dat;
model;
option solver ipopt;
option show_stats 1;
```

```
option ipopt_options '\
dual_inf_tol=1e-6
max_iter=5000
' :
option opt2 $ipopt_options ' warm_start_init_point=yes';
# NCL method.
# kmax, rhok, rhofac, rhomax, etak, etafac, etamin, rtol
# are defined in the .mod file.
printf "NCLipopt log for pTax5D\n" > 5DNCLipopt.log;
display na, nb, nc, nd, ne, primreg > 5DNCLipopt.log;
printf " k
                 rhok
                            etak
                                    rnorm
                                                 Obj\n" > 5DNCLipopt.log;
for {K in 1..kmax}
{ display na, nb, nc, nd, ne, primreg, K, kmax, rhok, etak;
  if K == 2 then {option ipopt_options $opt2 ' mu_init=le-4'};
   if K == 4 then {option ipopt_options $opt2 ' mu_init=1e-5'};
  if K == 6 then {option ipopt_options $opt2 ' mu_init=le-6'};
   if K == 8 then {option ipopt_options $opt2 ' mu_init=1e-7'};
   if K ==10 then {option ipopt_options $opt2 ' mu_init=1e-8'};
   display $ipopt_options;
   solve;
   let rmax := max({(i,j,k,g,h) in T, (p,q,r,s,t) in T:
     !(i=p and j=q and k=r and g=s and h=t)} R[i,j,k,g,h,p,q,r,s,t]);
   let rmin := min({(i,j,k,g,h) in T, (p,q,r,s,t) in T:
     !(i=p and j=q and k=r and g=s and h=t)} R[i,j,k,g,h,p,q,r,s,t]);
   display na, nb, nc, nd, ne, primreg, K, rhok, etak, kmax;
   display K, kmax, rmax, rmin;
   let rnorm := max(abs(rmax), abs(rmin)); # ||r||_inf
   printf "%4i %9.1e %9.1e %9.1e %15.7e\n", K, rhok, etak, rnorm, f >> 5DNCLipopt.log;
  close 5DNCLipopt.log;
   if rnorm <= rtol then
   { printf "Stopping: rnorm is small\n"; display K, rnorm; break; }
   if rnorm <= etak then # update dual estimate dk; save new solution
   {let {(i,j,k,g,h) in T, (p,q,r,s,t) in T:
         !(i=p and j=q and k=r and g=s and h=t)}
            dk[i,j,k,g,h,p,q,r,s,t] :=
            dk[i,j,k,q,h,p,q,r,s,t] + rhok \times R[i,j,k,q,h,p,q,r,s,t];
    let {(i,j,k,g,h) in T} ck[i,j,k,g,h] := c[i,j,k,g,h];
    let {(i,j,k,g,h) in T} yk[i,j,k,g,h] := y[i,j,k,g,h];
    display K, etak;
    if etak == etamin then { printf "Stopping: etak = etamin\n"; break; }
    let etak := max(etak*etafac, etamin);
   display etak;
   }
   else # keep previous solution; increase rhok
```

```
{ let {(i,j,k,g,h) in T} c[i,j,k,g,h] := ck[i,j,k,g,h];
     let {(i,j,k,g,h) in T} y[i,j,k,g,h] := yk[i,j,k,g,h];
     display K, rhok;
     if rhok == rhomax then { printf "Stopping: rhok = rhomax\n"; break; }
     let rhok := min(rhok*rhofac, rhomax);
    display rhok;
   }
}
display c,y; display na, nb, nc, nd, ne, primreg, rhok, etak, rnorm;
# Count how many constraint are close to being active.
data;
let nT
       := na*nb*nc*nd*ne; let m := nT*(nT-1); let n := 2*nT;
let etak := 1.0001e-10;
printf "\n m = %8i\n n = %8i\n", m, n >> 5DNCLipopt.log;
printf "\n Constraints within tol of being active\n\n" >> 5DNCLipopt.log;
printf " tol
                   count count/n\n" >> 5DNCLipopt.log;
for {K in 1..10}
{
 let kmax := card{(i,j,k,g,h) in T, (p,q,r,s,t) in T:
                  !(i=p and j=q and k=r and g=s and h=t)
                  and Incentive[i,j,k,g,h,p,q,r,s,t].slack <= etak};</pre>
 printf "%9.1e %8i %8.1f\n", etak, kmax, kmax/n >> 5DNCLipopt.log;
let etak := etak*10;
}
printf "Created 5DNCLipopt.log\n";
```